

```

FUNCTION z_demo_shelp_exit_add_fields.
*-----
*""Local Interface:
*  TABLES
*      SHLP_TAB TYPE SHLP_DESCR
*      RECORD_TAB STRUCTURE SEAHLPRES
*  CHANGING
*      VALUE(SHLP) TYPE SHLP_DESCR
*      VALUE(CALLCONTROL) LIKE DDSHF4CTRL STRUCTURE DDSHF4CTRL
*-----

DATA: lo_helper          TYPE REF TO zcl_shlp_demo_helper,
      lt_shlp_tab_temp   TYPE zcl_shlp_demo_helper=>gtt_shlp_descr,
      lt_record_tab_temp TYPE lxe_tt_seahlpres.

IF callcontrol-step = 'PRESEL'.
  CREATE OBJECT lo_helper.

*- Check if fields have been already been added
  READ TABLE shlp-interface TRANSPORTING NO FIELDS
    WITH KEY shlpfield = 'F_MOVIE'.
  IF sy-subrc = 0.
    EXIT.
  ENDIF.

*- Append fields to output structure
  CALL METHOD lo_helper->add_fields_to_shlp_interface
    CHANGING
      ch_interface = shlp-interface.

  CALL METHOD lo_helper->add_fields_to_shlp_fielddescr
    CHANGING
      ch_fielddescr = shlp-fielddescr.

  CALL METHOD lo_helper->add_fields_to_shlp_fieldprop
    CHANGING
      ch_fieldprop = shlp-fieldprop.
  ENDIF.

IF callcontrol-step = 'DISP'.
  CREATE OBJECT lo_helper.

*- Moving to temp structures so they can be used as method parameters
* (avoid OCCURS 0 in function module TABLES statements)
  lt_shlp_tab_temp = shlp_tab[].
  lt_record_tab_temp = record_tab[].

  CALL METHOD lo_helper->fill_customer_favorites_info
    EXPORTING
      in_shlp_tab   = lt_shlp_tab_temp
    CHANGING
      ch_record_tab = lt_record_tab_temp
      ch_shlp       = shlp
      ch_callcontrol = callcontrol.

*- Copy results back to RECORD_TAB

```

```
    record_tab[] = lt_record_tab_temp[].  
ENDIF.  
  
ENDFUNCTION.
```